10

CAMPOS VECTORIALES

10.1. Campos Vectoriales

Función VectorPlot[ ]

★ **VectorPlot**[{v_x, v_y}, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}]

Dibuja el campo de vectores {y , x} en cada punto del plano

```
campvec = VectorPlot[{y, x}, {x, -3, 3}, {y, -3, 3},
  VectorStyle → RGBColor[0.5, 0, 0.7], VectorScale → Small]
```



```
VectorPlot[{{-x, y}, {y, x}}, {x, -3, 3}, {y, -3, 3}]
```



▼ Opciones de la Función VectorPlot[ ]

★ VectorColor y VectorScale

```
campovectorial = VectorPlot[{y, x}, {x, -1, 1},
  {y, -1, 1}, VectorScale → Medium, VectorColorFunction → "Rainbow"]
```



★ Stream

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, StreamPoints → 15, StreamStyle → Purple,
 StreamScale → Full, VectorPoints → 12, VectorScale → Medium, VectorStyle → Automatic]
```



★ Regiones

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, RegionFunction → Function[{x, y}, x y < 0]]
```



★ **Campos de vectores en 3D**

```
VectorPlot3D[{x, y, z}, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```



10.2. Soluciones de una Ecuación Diferencial Ordinaria

▼ DSolve[ecuación, función, variable]

Resuelve una E.D. para una función y(x)

★ **Resolver una E.D.O. de orden 1**

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4\, y[x] + y'[x] == 0$

```
S1 = DSolve[ed1, y[x], x]
```

$\{\{y[x] \to e^{-4x}\, C[1]\}\}$

★ **Resolver una E.D.O. de 2º orden**

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2\, y[x] - 3\, y'[x] + y''[x] == 0$

```
S2 = DSolve[ed2, y[x], x]
```

$\{\{y[x] \to e^{x}\, C[1] + e^{2x}\, C[2]\}\}$

★ **NDsolve[ecuación, función, $\{x, x_{min}, x_{max}\}$]**

Resuelve una E.D. de forma numérica en el intervalo dado

```
s = NDSolve[{y'[x] == y[x] Cos[x + y[x]], y[0] == 1}, y, {x, 0, 30}]
{{y → InterpolatingFunction[{{0., 30.}}, <>]}}
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
```



10.3. Ecuaciones Diferenciales Ordinarias de primer orden

▼ Solución General y Solución Particular

★ Solución general de una E.D.O.

Definimos una función que sea la solución general de la E.D.O. de orden 1 dada, a partir de la solución obtenida. Es una familia uniparamétrica de soluciones

```
sg[x_, c_] = S1[[1, 1, 2]] /. C[1] → c
```

$c\, e^{-4x}$

★ Familia de soluciones

Generamos una familia de soluciones dando valores al parámetro c

```
listsol = Table[sg[x, c], {c, -4, 4, 1}]
```

$\{-4 e^{-4x}, -3 e^{-4x}, -2 e^{-4x}, -e^{-4x}, 0, e^{-4x}, 2 e^{-4x}, 3 e^{-4x}, 4 e^{-4x}\}$

```
famsol =
 Plot[Evaluate[listsol], {x, 0, 1}, PlotStyle → Thickness[0.005], PlotRange → {-4, 4}]
```



★ Solución de una E.D.O. con condiciones iniciales

Hallar la solución de la ecuación ed1 que verifica y(0)=2

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4\,y[x] + y'[x] == 0$

```
solg = DSolve[{ed1, y[x0] == y0}, y[x], x]
```

$\{\{y[x] \to e^{-4x+4x0}\,y0\}\}$

```
yg[x_] = solg[[1, 1, 2]]
```

$e^{-4x+4x0}\,y0$

```
yg[x] /. {x0 → 0, y0 → 2}
```

$2\,e^{-4x}$

```
sp = DSolve[{ed1, y[0] == 2}, y[x], x]
```

$\{\{y[x] \to 2\,e^{-4x}\}\}$

```
yp[x_] = sp[[1, 1, 2]]
```

$2\,e^{-4x}$

Hallar la solución de la ecuación de 0rden 2, que verifica y(0)=2, y’(0)=1

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2\,y[x] - 3\,y'[x] + y''[x] == 0$

```
sp2 = DSolve[{ed2, y[0] == 2, y'[0] == 1}, y[x], x]
```

$\{\{y[x] \to -e^{x}\,(-3 + e^{x})\}\}$

```
yp2[x_] = sp2[[1, 1, 2]]
```

$-e^{x}\,(-3 + e^{x})$

Particularidaes.

Para la E.D.O. x^2*y’[x]+y[x]^2==0, creamos una familia de soluciones a partir de la solución general

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\,C[1]}\right\}\right\}$$

```
solgen[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$$-\frac{x}{1 + c\,x}$$

```
listsol = Table[solgen[c, x], {c, -2, 2, .5}]
```

$$\left\{-\frac{x}{1 - 2.\,x}, -\frac{x}{1 - 1.5\,x}, -\frac{x}{1 - 1.\,x}, -\frac{x}{1 - 0.5\,x}, -\frac{x}{1 + 0.\,x}, -\frac{x}{1 + 0.5\,x}, -\frac{x}{1 + 1.\,x}, -\frac{x}{1 + 1.5\,x}, -\frac{x}{1 + 2.\,x}\right\}$$

```
familiasol = Plot[Evaluate[listsol], {x, -2, 3}]
```



Si queremos resolver el problema de valores iniciales {x^2*y'[x]+y[x]^2==0,y[0]==2} con DSolve no obtenemos ninguna solución.

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 2}, y[x], x]
```

DSolve::bvnul :
For some branches of the general solution, the given boundary conditions lead to an empty solution. ≫

{}

Lo intentamos, hallando primero la solución general de la ecuación y después el valor de c para la condición inicial dada. Seguimos sin obtener ninguna solución. De hecho, en la gráfica puede observarse que por el punto (0,2) no pasa ninguna solución.

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\,C[1]}\right\}\right\}$$

```
solgen[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$$-\frac{x}{1 + c\,x}$$

```
cons = Solve[solgen[c, 0] == 2, c]
```

{}

Si resolvemos el problema de valores iniciales{x^2*y'[x]+y[x]^2==0,y[0]==0} con DSolve, nos dan infinitas soluciones. En la gráfica puede observarse que hay muchas soluciones que pasan por el punto (0,0).

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 0}, y[x], x]
```

DSolve::bvnr :
For some branches of the general solution, the given boundary conditions do not restrict the existing freedom in the general solution. ≫

DSolve::bvsing :
Unable to resolve some of the arbitrary constants in the general solution using the given boundary conditions. It is possible that some of the conditions have been specified at a singular point for the equation. ≫

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\,C[1]}\right\}\right\}$$

10.4. Campo vectorial asociado a una E.D.O. de orden 1

▼ Campo de tangentes

El campo de vectores tangentes asociado a la Ecuación Diferencial y'=f(x,y), vendrá vendrá dado por {1,f(x,y)}

★ **Dada la ecuación diferencial y'=x, resolvemos y definimos la solución general y dibujamos la familia de soluciones**

```
ed = y'[x] == x
```

```
y′[x] == x
```

```
S = DSolve[ed, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{x^2}{2} + C[1]\right\}\right\}$$

```
sg[x_, c_] = S[[1, 1, 2]] /. C[1] → c
```

$$c + \frac{x^2}{2}$$

```
listsol = Table[sg[x, c], {c, -5, 8, 1}]
```

$$\left\{-5 + \frac{x^2}{2}, -4 + \frac{x^2}{2}, -3 + \frac{x^2}{2}, -2 + \frac{x^2}{2}, -1 + \frac{x^2}{2}, \frac{x^2}{2}, 1 + \frac{x^2}{2}, 2 + \frac{x^2}{2}, 3 + \frac{x^2}{2}, 4 + \frac{x^2}{2}, 5 + \frac{x^2}{2}, 6 + \frac{x^2}{2}, 7 + \frac{x^2}{2}, 8 + \frac{x^2}{2}\right\}$$

```
famsol = Plot[Evaluate[listsol], {x, -5, 5},
  PlotStyle → {Green, Thickness[0.005]}, PlotRange → {-5, 8}]
```



★ **Función VectorPlot** [{ v_x, v_y }, { x, x_{min}, x_{max} }, { y, y_{min}, y_{max} }]

El campo de vectores tangentes asociado a la Ecuación Diferencial y'=x, vendrá dado por el campo vectorial {1,x}

```
campvec = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorStyle → RGBColor[0.2, 0.8, 0.3], VectorScale → Small]
```



★ Juntando las gráficas anteriores

```
Show[{campvec, famsol}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ **Opción “stream”**

```
campvec = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, StreamScale → Full, StreamPoints → 15, StreamScale → Full]
```



▼ Trayectorias Ortogonales

Las trayectorias ortogonales de la familia de curvas solución de la ecuación diferencial y’ = x, tendrán por ecuación diferencial y’ = -1/x y por vector tangente el vector m’= {- f (x, y).1} = {- x,1}

★ **Resolvemos la E.D. de la trayectorias ortogonales y definimos la solución general y dibujamos la familia de curvas**

```
edto = y'[x] == -1 / x
```

$$y'[x] == -\frac{1}{x}$$

```
Sto = DSolve[edto, y[x], x]
```

```
{{y[x] → C[1] - Log[x]}}
```

```
sgto[x_, c_] = Sto[[1, 1, 2]] /. C[1] → c
```

```
c - Log[x]
```

```
listsol1 = Table[sgto[x, c], {c, -5, 8, 1}]
```

```
{-5 - Log[x], -4 - Log[x], -3 - Log[x], -2 - Log[x], -1 - Log[x], -Log[x], 1 - Log[x],
 2 - Log[x], 3 - Log[x], 4 - Log[x], 5 - Log[x], 6 - Log[x], 7 - Log[x], 8 - Log[x]}
```

```
listsol2 = Table[sgto[-x, c], {c, -5, 8, 1}]
```

```
{-5 - Log[-x], -4 - Log[-x], -3 - Log[-x], -2 - Log[-x], -1 - Log[-x], -Log[-x], 1 - Log[-x],
 2 - Log[-x], 3 - Log[-x], 4 - Log[-x], 5 - Log[-x], 6 - Log[-x], 7 - Log[-x], 8 - Log[-x]}
```

```
famsol1 = Plot[Evaluate[listsol1], {x, 0.01, 3},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
```

```
famsol2 = Plot[Evaluate[listsol2], {x, -3, -0.01},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
```

```
famsolto = Show[{famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



Si el campo de vectores asociado a la Ecuación Diferencial y'=x, tiene por vector tangente el vector {1,f(x,y)}={1,x}, el de las trayectorias ortogonales será {-x,1}

```
g1 = VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, VectorPoints → 12, VectorStyle → {Purple, Blue}]
```




```
g2 = Show[{famsol, famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ Dibujamos conjuntamente las familias de curvas y los campos de tangentes

```
Show[{g1, g2}, PlotRange → {{-3, 3}, {-3, 8}}]
```




```
VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8}, StreamPoints → 15, StreamScale → Full,
 VectorPoints → 12, VectorScale → Small, VectorStyle → {Purple, Blue}]
```